ORNL-4179
UC-34 – Physics

A FORTRAN PROGRAM FOR RELATIVISTIC KINEMATIC CALCULATIONS IN TWO-BODY NUCLEAR REACTIONS

R. E. Phillips
S. T. Thornton

OAK RIDGE NATIONAL LABORATORY
operated by
UNION CARBIDE CORPORATION
for the
U.S. ATOMIC ENERGY COMMISSION

Printed in the United States of America. Available from Clearinghouse for Federal
Scientific and Technical Information, National Bureau of Standards,
U.S. Department of Commerce, Springfield, Virginia 22151
Price: Printed Copy $3.00; Microfiche $0.65

ORNL-4179

Contract No. W-7405-eng-26

PHYSICS DIVISION

A FORTRAN PROGRAM FOR RELATIVISTIC KINEMATIC CALCULATIONS IN TWO-BODY NUCLEAR REACTIONS

R. E. Phillips and S. T. Thornton

NOVEMBER 1967

OAK RIDGE NATIONAL LABORATORY
Oak Ridge, Tennessee
operated by
UNION CARBIDE CORPORATION
for the
U.S. ATOMIC ENERGY COMMISSION

TABLE OF CONTENTS

ABSTRACT

We have developed a computer program for calculating the kinematics of two-body nuclear reactions using relativistic formulas. The energies and scattering angles of the two decay particles in the laboratory system and the energies of all four particles in the center-of-mass system, along with the center-of-mass to laboratory differential cross-section ratio are calculated. The input consists of the four masses, the Q-value, the laboratory scattering angle, and the initial laboratory bombarding energy. The program exists for both the Control Data 1604-A and IBM System/360 Model 75 computers in both regular and double precision FORTRAN.

I. Introduction

Relativistic kinematic relations are often needed for two-body nuclear reaction calculations. This report contains a brief summary of certain relativistic equations and a description of a computer program which employs these equations to compute the kinematic relations for such reactions.

All of the equations used in the following discussion and in the relativistic kinematic computer program RELKIN are taken from one of the following sources: (1) An Introduction to Elementary Particles by W. S. C. Williams[1]; (2) Kinematics of Nuclear Reactions by A. M. Baldin, V. I. Gol'danskii, and I. L. Rozenthal[2]; and (3) Classical Electrodynamics by John David Jackson[3].

Consider the reaction

$$1 + 2 \rightarrow 3 + 4 \qquad (1)$$

in which 1 is the projectile particle, 2 is the target, 3 is the outgoing particle, and 4 is the recoiling nucleus.

The reaction (1) is considered in both the laboratory and center-of-mass systems. The primed and unprimed symbols represent the center-of-mass and laboratory systems, respectively. The particles and scattering angles are shown in Fig. 1 for both systems. The symbols used are:

a) p = momentum;

b) m = rest mass;

c) E = total energy of a particle;

d) W = kinetic energy;

e) θ = scattering angle.



Fig. 1. Schematic Representation of a Two-Body Nuclear Reaction

It is assumed that the initial kinetic energy of particle 1, the rest masses of all four particles, the Q-value, and the laboratory scattering angle of particle 3 are known. From these initial values the other kinematic factors are to be calculated.

II. Kinematic Relationships

The total energy of a particle is

$$E = W + mc^2, \tag{2}$$

and

$$E^2 = p^2c^2 + m^2c^4. \tag{3}$$

We will use a system in which c = 1. Therefore

$$E = W + m, \tag{4}$$

and

$$E^2 = p^2 + m^2. \tag{5}$$

The kinetic energy of the projectile particle is W_1, so that

$$E_1 = W_1 + m_1, \tag{6}$$

and from Eq. (5)

$$p_1 = (E_1^2 - m_1^2)^{1/2}. \tag{7}$$

The Q-value for a two-body reaction is defined by

$$Q = m_1 + m_2 - (m_3 + m_4). \tag{8}$$

The minimum kinetic energy required of particle 1 for the reaction to

proceed is the threshold kinetic energy, W_{TH}, where

$$W_{TH} = \frac{-Q}{2\, m_2} (m_1 + m_2 + m_3 + m_4). \tag{9}$$

Thus the minimum total energy for particle 1 required for the reaction to proceed is, from Eq. (6),

$$E_{TH} = W_{TH} + m_1. \tag{10}$$

The total energy, E_T, of the system is defined by

$$E_T = E_1 + E_2 = E_3 + E_4, \tag{11a}$$

and therefore,

$$E_T = E_1 + m_2. \tag{11b}$$

In the center-of-mass system, where the total momentum is defined as zero, we have

$$E_T' = (m_1^2 + m_2^2 + 2\, m_2\, E_1)^{1/2}, \tag{12}$$

$$p_1' = p_2' = \frac{p_1 m_2}{E_T'}, \tag{13}$$

$$E_1' = \frac{m_1^2 + m_2 E_1}{E_T'}, \tag{14}$$

and

$$E_2' = \frac{m_2^2 + m_2 E_1}{E_T'}. \tag{15}$$

For the outgoing particles we have

$$E_T' = E_3' + E_4' \ , \tag{16}$$

$$E_3' = \frac{E_T'^2 + m_3^2 - m_4^2}{2\,E_T'} \ , \tag{17}$$

$$E_4' = \frac{E_T'^2 + m_4^2 - m_3^2}{2\,E_T'} \ , \tag{18}$$

$$p_3' = (E_3'^2 - m_3^2)^{1/2} \ , \tag{19}$$

$$p_4' = (E_4'^2 - m_4^2)^{1/2} \ , \tag{20}$$

and

$$p_3' = p_4' \ . \tag{21}$$

The laboratory scattering energy of particle 3 is

$$E_3 = \frac{1}{E_T^2 - p_1^2\cos^2\theta_3} \left\{ E_T \left(m_2 E_1 + \frac{m_1^2 + m_2^2 + m_3^2 - m_4^2}{2} \right) \right.$$

$$\pm p_1 \cos\theta_3 \left[\left(m_2 E_1 + \frac{m_1^2 + m_2^2 - m_3^2 - m_4^2}{2} \right)^2 - m_3^2 m_4^2 \right.$$

$$\left. \left. - p_1^2 m_3^2 \sin^2\theta_3 \right]^{1/2} \right\} \ . \tag{22}$$

The ambiguity of the ± sign is cleared up by the quantity α defined by

$$\alpha = \frac{p_1}{E_T} \frac{1 + \frac{m_3^2 - m_4^2}{E_T'^2}}{\left\{\left[1 - \left(\frac{m_3 + m_4}{E_T'}\right)^2\right]\left[1 - \left(\frac{m_3 - m_4}{E_T'}\right)^2\right]\right\}^{1/2}} . \tag{23}$$

When $\alpha > 1$, both roots in Eq. (22) are allowed and two solutions exist for the scattering angle θ_3. However, if $\alpha < 1$, only one solution is possible, and the positive sign is chosen. Obviously the term within the square root in Eq. (22) must not be negative. This fact is used to define the maximum laboratory scattering angles of particles 3 and 4.

$$\sin^2\theta_3 \leq \frac{(E_T^2 - p_1^2 + m_3^2 - m_4^2)^2 - 4 m_3^2 [(m_1 + m_2)^2 + 2 m_2 W_1]}{4 m_3^2 p_1^2} , \tag{24}$$

and

$$\sin^2\theta_4 \leq \frac{(E_T^2 - p_1^2 + m_4^2 - m_3^2)^2 - 4 m_4^2 [(m_1 + m_2)^2 + 2 m_2 W_1]}{4 m_4^2 p_1^2} . \tag{25}$$

The laboratory energy of particle 4 is then

$$E_4 = E_T - E_3 . \tag{26}$$

The momenta of particles 3 and 4 are now easily determined from Eq. (5).

$$p_3 = (E_3^2 - m_3^2)^{1/2} , \tag{27}$$

and

$$p_4 = (E_4^2 - m_4^2)^{1/2} . \tag{28}$$

The scattering angles θ_3', θ_4', and θ_4 can now be found. θ_3' may be determined by

$$\sin\theta_3' = \frac{p_3}{p_3'} \sin\theta_3, \tag{29}$$

and

$$\cos\theta_3' = \frac{E_T}{E_T' p_3'} \left[p_3 \cos\theta_3 - \frac{p_1 E_3}{E_T} \right] . \tag{30}$$

The scattering angle of θ_4' is, of course,

$$\theta_4' = 180° - \theta_3'. \tag{31}$$

Finally, the lab scattering angle, θ_4, can be found from

$$\sin\theta_4 = \frac{p_3}{p_4} \sin\theta_3, \tag{32}$$

and

$$\cos\theta_4 = \frac{p_1 - p_3 \cos\theta_3}{p_4} . \tag{33}$$

The final required relation is the ratio of the center-of-mass differential cross section to the laboratory differential cross section. It is given by

$$\frac{\left(\frac{d\sigma}{d\Omega} (\theta_3) \right)'}{\left(\frac{d\sigma}{d\Omega} (\theta_3) \right)} = \frac{E_T}{E_T'} \left(\frac{\sin\theta_3}{\sin\theta_3'} \right)^3 \left\{ 1 + \cos\theta_3' \frac{p_1}{E_T} \frac{E_T^2 - p_1^2 + m_3^2 - m_4^2}{\left[(E_T^2 - p_1^2 + m_3^2 - m_4^2)^2 - 4 E_T'^2 m_3^2 \right]^{1/2}} \right\} . \tag{34}$$

III. Computer Program

A. Program RELKIN

The program RELKIN has been written for both the CDC-1604-A computer (FORTRAN-63) and the IBM System/360, Model 75 computer (FORTRAN-IV). Both single precision and double precision versions are available for each computer. The single precision version for the IBM-360 computer is only accurate to about three figures and the double precision version should be used. Only standard input and output magnetic tapes are used.

The estimated time for the IBM-360 computer to perform one complete kinematic calculation for one input energy and one angle is .017 sec in double precision.

The following sections describe the computer program and the input and output information. Ample comment cards exist in the program to help in understanding the various options and procedures.

A listing of the computer program written in double precision (FORTRAN-IV) may be found in the Appendix.

B. Description

Initially the computer reads three data cards. From these cards the computer gets values for the masses of the incident particle, the target, the detected particle, and the recoil nucleus. In addition, three arrays, Q(I), EINI(I), and THETA(I), are filled. The first array may contain up to four Q-values. EINI(I) contains values of the initial laboratory bombarding kinetic energy and THETA(I) contains the laboratory scattering angles of the detected particle. Each of the last two arrays may contain up to ten values.

An option allows the user to give values to the first, second, and tenth elements of EINI(I). This causes the program to compute the desired results for values of EINI(I) from EINI(1) to EINI(2) by increments of EINI(10). The same option is available for THETA(I), but not for Q(I). See Section III-C for input formats and units.

Program RELKIN is basically a nest of three do-loops. The outermost loop chooses Q-values. For each Q-value the second loop chooses the different initial lab bombarding energies (EINI). Finally, for each bombarding energy the inner-most do-loop chooses values of the laboratory scattering angles of the detected particle (THETA) from which the final results are calculated and tabulated.

To create a more flexible program, preceding the middle, or EINI, do-loop there is an "increment check." If the computer finds non-zero values for only the first, second, and tenth elements of the array EINI(I), it will compute final results using values of EINI(I) from EINI(1) to EINI(2), incrementing by steps of EINI(10). Since this check occurs just before the second do-loop, it is a simple procedure to exit the do-loop after one run, increment EINI(1) by EINI(10), and begin the do-loop again. Of course, if the conditions for incrementing are not fulfilled, the do-loop is not exited until either a zero-value or the tenth value of EINI(I) is reached. This same increment-checking procedure is used for the inner-most do-loop (THETA).

Besides routine tests for zero and non-zero values of data, there are three checks to insure that the data is within certain bounds. The first operation which occurs within the outside, Q, do-loop is a test of the Q-value. The Q-value should be equal to or less than $m_1 + m_2 - m_3 - m_4$ (Eq. (8), ground state Q value). If the read-in Q-value is greater than

this ground state Q-value by .01 MeV or more, an error message is given. No further computations are initiated with this particular read-in value. A new Q-value is selected and the program begins again. If a Q-value is read in that is lower than the ground state Q-value, it is assumed that the reaction proceeds to an excited state of the residual nucleus, particle 4. The difference between the ground state Q-value and the read-in Q-value is then added to the mass of particle 4.

The second check is a comparison of the total energy of the bombarding particle, Eq. (6), and the total threshold energy, Eq. (10). If the energy of the bombarding particle is not greater than the threshold energy, an error message is generated. The program then selects the next bombarding energy and the sequence begins again. A new Q-value is not selected until all the incident energies and their associated angles have been processed.

A final check determines if the incident energy is sufficient to initiate a reaction at the given laboratory scattering angle THETA. For the reaction to proceed the restrictions of Eq. (24) and Eq. (25) must be satisfied. If these limitations are not adhered to, the same results occur as in the preceding check. The program does <u>not</u> go to the next angle, but to a new bombarding energy. Thus it is imperative that angles be placed in increasing order of value on the data card (obviously, this eliminates incrementing the angle THETA by a negative quantity).

As mentioned in the preceding section, a set of data may yield two solutions if the variable α is greater than 1 (Eq. 23). A simple do-loop (I = 1, IJ), where IJ = 2 if $\alpha > 1$, takes this possibility into account.

Other than the preceding exceptions, the program is a straightforward, simple (if tedious) set of calculations.

C. Input Data

Card	Variables	Description	Units	Format
1	XM1A, XM2A, XM3A, XM4A	m_1, m_2, m_3, m_4; particle masses	amu in ^{12}C system	4F12.8
	Q(I)	Q-values, up to 4 values	MeV	4F8.4
2	EINI(I)	laboratory bombarding kinetic energy of particle 1, increment or up to 10 values (see Section III-B)	MeV	10F8.4
3	THETA(I)	laboratory scattering angle of particle 3, increment or up to 10 values (see Section III-B).	degrees	10F8.3

There is no limit on the number of cases which can be run at one time. Each case must contain only these cards in this order.

D. Output Data

The printout includes:

1) The four input masses (^{12}C amu) and the input Q-values (MeV);

2) The laboratory bombarding energy (MeV) and the center-of-mass energies for all four particles (MeV);

3) For the two solutions (if applicable):

   a. the laboratory scattering angle of particle 3 (degrees);

b. center-of-mass to laboratory ratio of differential cross sections of particle 3;

c. laboratory kinetic energies of particles 3 and 4 (MeV);

d. center-of-mass angle θ_3 (degrees);

e. laboratory angle θ_4 (degrees).

E. Variable Definitions for Program RELKIN

Program Names	Section II Names
XM1A, XM2A, XM3A, XM4A	m_1, m_2, m_3, m_4; masses of particles $1 \rightarrow 4$, respectively, in amu (^{12}C system)
Q(I)	Q-values, maximum of 4, in MeV
EINI(I)	W_1, laboratory bombarding energies, maximum of 10, unless incremented, in MeV
THETA(I)	θ_3, read in as degrees, maximum of 10 unless incremented
QGS	ground state Q-value, MeV (see Eq. (8))
XM4	m_4 + (QGS-Q)
E1, ETH	E_1, E_{TH}
ET	E_T
P1	p_1
SITH3, CØTH3	$\sin\theta_3$, $\cos\theta_3$
ECMT	E'_T
ECM1, ECM2, ECM3, ECM4	E'_1, E'_2, E'_3, E'_4
PCM	p'_3, p'_4
ALPHA	α

E3(I), E4(I)	E_3, E_4
P3, P4	p_3, p_4
CØCM, SICM	$\cos\theta'_3$, $\sin\theta'_3$
THCM3(I), THCM4(I)	θ'_3, θ'_4
SITH4, CØTH4	$\sin\theta_4$, $\cos\theta_4$
THETA4(I)	θ_4
CMTØLB	ratio of center-of-mass to laboratory differential cross sections (see Eq. (34))

F. Obtaining Program Decks

FORTRAN decks for this program may be obtained by writing to the address below. It would be appreciated if any faults or errors found in the program were reported.

S. T. Thornton
Department of Physics
University of Wisconsin
Madison, Wisconsin 53706

Appendix

This appendix contains a listing of the program RELKIN as it was written for the IBM-360/75 computer at Oak Ridge. It is in double precision. Since the single and double precision versions of the program written for the CDC-1604-A are so similar to the IBM-360/75 version, their listings are not included within the report.

```
C      PROGRAMMED BY  R.E. PHILLIPS AND S. T. THORNTON
C      OAK RIDGE NATIONAL LAB, JULY, 1967
C      SEE ORNL REPORT 4179
C
       PROGRAM RELKIN
C
C      USE DOUBLE PRECISION
       DOUBLE PRECISION XM1A, XM2A, XM3A, XM4A, Q, EINI, THETA, EN,
      1 XM1, XM2, XM3, XM4B, QGS, XM4, T, E1, ETH, ET, P1, TA,
      2SITH3, COTH3, CHECK, ECMT, ECM1, ECM2, ECM3, ECM4, PCM, ALPHA,
      3E3, E4, P3, P4, BETA, GAMMA, COCM, SICM, THCM3, THCM4, SITH4,
      4COTH4, THETA4,          CMTOLB
C
       DIMENSION  CMTOLB(2), E3(2), THCM3(2), E4(2), THCM4(2),THETA4(2)
      1, THETA(10), EINI(10) ,Q(4)
C
C      FIRST DATA CARD CONTAINS MASSES (4F12.8) AND Q VALUES(4F8.4)
C      SECOND DATA CARD CONTAINS BOMBARDING ENERGIES (10F8.4). E OPTION
C      THIRD  DATA CARD CONTAINS LAB SCATTERING ANGLES(10F8.3).   OPTION
C
C      INPUT MASSES IN C12 AMU UNITS
C      INPUT LAB ANGLES IN DEGREES
C      INPUT Q VALUES AND ENERGIES IN MEV
C
C      READ ALL THREE DATA CARDS
   50  READ51, XM1A,XM2A,XM3A,XM4A,Q, EINI, THETA
   51  FORMAT (4D12.8,4D8.4/10D8.4/10D8.3)
C
       EN = EINI(1)
       T = THETA (1)
C
C      CHANGE REST MASSES TO ENERGY (MEV)
       XM1 = XM1A* 931.441
       XM2 = XM2A* 931.441
       XM3 = XM3A* 931.441
       XM4B = XM4A* 931.441
C
C       THIS DO LOOP COMPUTES FOR EACH Q VALUE READ IN
       DO 13 MQ = 1, 4
       IF (DABS(Q(MQ))) 6,6,12
```

```
C
C       QGS  GROUND STATE Q VALUE
  12    QGS = XM1 + XM2- XM3 - XM4B
C
C        THIS IS A CHECK TO INSURE REASONABLE Q VALUES AND MASSES
C       INPUT Q SHOULD BE LESS THAN OR EQUAL TO QGS
        IF (.01 - (Q(MQ) -QGS)) 30, 30, 31
  30    PRINT52
  52    FORMAT(/////41H EITHER MASS VALUES OR Q VALUE WAS BOOTED)
        GO TO 13
C
C       IF Q IS LESS THAN QGS, IT IS ASSUMED THAT RESIDUAL NUCLEUS (PARTIC
C       LE 4)  IS IN EXCITED STATE OF QGS-Q . ADD THIS ENERGY TO MASS.
  31    XM4 = XM4B + QGS - Q(MQ)
C
C       PRINT PART OF THR OUTPUT DATA
        PRINT56, XM1A, XM2A, XM3A, XM4A, Q(MQ)
  56    FORMAT (1H1,13H INCIDENT M =,F9.5, 12H  TARGET M =, F9.5,  14H  DE
       1TECTED M =,F9.5,  12H  RECOIL M =,F9.5, 17H (AMU)  Q VALUE =, F9.4
       2,4H MEV)
C
        EINI(1) = EN
C
C        NEXT TWO STATEMENTS DETERMINE IF ENERGIES ARE READ IN OR DONE
C       BY INCREMENTS
        IF (EINI(3) - 1.E -6) 20, 20, 21
  20    IF (EINI(10) - 1.E- 6) 21, 21, 22
C
  22    NOT = 2
C       ENERGY DONE BY INCREMENTS
        GO TO 24
C
  21    NOT = 1
C       ENERGY NOT DONE BY INCREMENTS
        GO TO 24
C
  23    EINI (1) = EINI(1) + EINI(10)
        IF (EINI(1) - EINI(2)) 24, 24, 13
```

```
C
C     THIS DO LOOP COMPUTES VALUES FOR EACH ENERGY
  24  DO 19 MO = 1, 10
      NS = 1
      IF (EINI(MO) - 1.E- 6) 13,13,26
C
C     DEFINE QUANTITIES TO BE USED IN LATER EQUATIONS
  26  E1 = EINI(MO) + XM1
C     CALCULATE THRESHOLD ENERGY
      ETH = - Q(MQ) * (XM1 + XM2 + XM3 + XM4) / (2. * XM2) + XM1
C
C     THE NEXT STATEMENT IS A CHECK FOR THRESHHOLD VALUES
      IF (E1 - ETH) 34, 34, 32
  34  PRINT53, EINI(MO)
  53  FORMAT(/25H LAB BOMBARDING ENERGY = ,F10.5,39H MEV IS BELOW REACTI
     1ON THRESHOLD ENERGY)
      GO TO 25
C
  32  ET = E1 + XM2
      P1 = DSQRT(E1 ** 2 - XM1 ** 2)
      THETA(1) = T
C
C     THE NEXT TWO STATEMENTS DETERMINE IF VALUES OF THETA 3 ARE READ IN
C     OR DONE BY INCREMENTS
      IF (THETA(3) - 1.E- 6) 14, 14, 15
   14 IF (THETA(10) - 1.E-6) 15, 15, 16
C
  16  NO = 2
C     ANGLES DONE BY INCREMENTS
      GO TO 18
C
  15  NO = 1
C     ANGLES NOT DONE BY INCREMENTS
      GO TO 18
C
  17  THETA (1) = THETA(1) + THETA (10)
      IF (THETA (1) - THETA (2)) 18, 18, 25
```

```
C
C     THIS DO LOOP COMPUTES VALUES FOR EACH ANGLE
  18  DO 33 L = 1, 10
      IF (THETA(L) - 1.E-6) 25, 25, 27
C
C     THE REST OF THE PROGRAM DOWN TO STATEMENT 5 COMPUTES THE ANGLES
C     AND ENERGIES
C
  27  TA = THETA(L)
C     CHANGE ANGLE TO RADIANS
      THETA(L) = THETA(L) * .017453292
      SITH3 = DSIN (THETA(L))
      COTH3 = DCOS (THETA(L))
C
C     CHECK TO SEE IF PARTICLE HAS ENOUGH ENERGY FOR ANGLE
      CHECK = (XM2 * E1 + (XM1 ** 2 + XM2 ** 2 - XM3 ** 2 - XM4 ** 2) /
     12.) ** 2 - (XM3 * XM4 ) ** 2 - (P1 * XM3 * SITH3) ** 2
      IF(CHECK)35, 36, 36
   35 PRINT  60, EINI(MO), TA
  60  FORMAT(25H LAB BOMBARDING ENERGY = ,F10.5,69H MEV IS ABOVE REACTIO
     1N THRESHOLD ENERGY, BUT TOO LOW FOR THETA LAB = ,F10.5)
      THETA(L) = TA
      GO TO 25
C
C     CALCULATE CENTER OF MASS ENERGIES
  36  ECMT = DSQRT(XM1 ** 2 + XM2 ** 2 + 2. *XM2 * E1)
      ECM1= (XM1 ** 2 + XM2 * E1) / ECMT
      ECM2= (XM2 ** 2 + XM2 * E1) / ECMT
      ECM 3 = ( ECMT ** 2 + XM3 ** 2 - XM4 ** 2) / (2. * ECMT )
      ECM 4 = ( ECMT ** 2 + XM4 ** 2 - XM3 ** 2)/ (2. * ECMT )
C
C     PCM = CENTER OF MASS MOMENTUM OF OUTGOING PARTICLES
      PCM  = DSQRT ( ECM4 ** 2 - XM4 ** 2)
C
C     ALPHA DETERMINES IF THERE IS ONE OR TWO SOLUTIONS FOR E3 AT ANGLE
C     THETA3
      ALPHA = P1 * (1. + (XM3 ** 2 - XM4 ** 2)/ ECMT ** 2) / (( ET
     1) * DSQRT((1. - ((XM3 + XM4) / ECMT) ** 2) * (1. - ((XM3 - XM4)/
     2ECMT) ** 2)))
```

```
C
      IF (ALPHA - 1.) 2, 2, 1
  2   IJ = 1
C     ONE SOLUTION FOR E3
      GO TO 4
C
  1   IJ = 2
C     TWO SOLUTIONS FOR E3
  4   DO 41 I = 1, IJ
C
C     CALCULATION OF ENERGY OF OUTGOING PARTICLE 3
      E3(I) = (ET * (XM2 * E1 + (XM1 ** 2 + XM2 ** 2 + XM3 ** 2 - XM4 **
     12) /2.) + (3 - 2 * I) * P1 * COTH3 * DSQRT(CHECK)) /(ET ** 2 - (P1
     1* COTH3) **2)
C
C     SINCE ET = E3 + E4
   7  E4(I) = ET - E3 (I)
C
C      CALCULATE MOMENTA OF PARTICLES 3 AND 4
      P3 = DSQRT (E3(I) ** 2 - XM3 ** 2)
      P4 = DSQRT ( E4(I) ** 2 - XM4 ** 2)
C
      BETA = P1 / ET
      GAMMA = ET / ECMT
C
C     CALCULATE CENTER OF MASS SCATTERING ANGLES OF PARTICLES 3 AND 4
      SICM = P3 * SITH3 / PCM
      COCM = GAMMA * (P3 * COTH3 - BETA * E3(I))/ PCM
      THCM3(I) = DATAN(SICM/ COCM)
      IF (COCM) 8, 9, 9
  8   THCM3(I) = 3.14159265 + THCM3(I)
   9  CONTINUE
      THCM4(I) = 3.14159265 - THCM3(I)
C
C     CALCULATE LAB SCATTERING ANGLE OF PARTICLE 4
      SITH4 = P3 * SITH3 / P4
      COTH4 = (P1 - P3 * COTH3) / P4
      THETA4(I) = DATAN ( SITH4 / COTH4)
      IF ( COTH4) 10, 11, 11
```

```
  10  THETA4(I) = 3.14159265 + THETA4(I)
  11  CONTINUE
C
C     DETERMINE CM TO LAB RATIO OF DIFFERENTIAL CROSS SECTION
      CM TO LB (I) = ET/ECMT*(1.+ (3.-2.*I) *
     1 P1/ET * (ET**2-P1**2+XM3**2-XM4**2)
     1/DSQRT((ET**2-P1**2+XM3**2-XM4**2)**2-4. *ECMT**2*XM3**2)
     2*COCM) *SITH3**3/ SICM**3
C
C     CHANGE ANGLES TO DEGREES
      THCM4(I) = THCM4 (I)/ .017453292
      THCM3(I) = THCM3 (I)/ .017453292
      THETA4(I) = THETA4(I)/ .017453292
C
C     CHANGE ENERGIES TO KINETIC (MEV)
      E3(I) = E3(I) - XM3
      E4(I) = E4(I) - XM4
  41  J = I
  5   ECM1 = ECM1 - XM1
      ECM2 = ECM2 - XM2
      ECM 3 = ECM 3 - XM3
      ECM 4 = ECM 4 - XM4
C
      THETA(L) = THETA(L)/ 0.017453292
C
      GO TO (28, 29),NS
C
C     PRINT OUTPUT
  28  PRINT57, EINI(MO),ECM1, ECM2, ECM3, ECM4
  57  FORMAT(///42X,25HLAB BOMBARDING ENERGY   =,F10.5,4H MEV//42X,25HCE
     1NTER OF MASS ENERGY 1 =,F10.5,/42X,25HCENTER OF MASS ENERGY 2 =,F1
     20.5/ 42X,25HCENTER OF MASS ENERGY 3 =,F10.5/ 42X,25HCENTER OF MASS
     3 ENERGY 4 =,F10.5//35X, 10HSOLUTION 1, 40X, 10HSOLUTION 2//
     3          6X,9HTHETA LAB,2(50H   CM TO LB  E3 LAB  THETA3 CM  E4 L
     1AB  THETA4 LAB)/ )
  29  PRINT 91,       TA,(CMTOLB(I), E3(I), THCM3(I), E4(I),THETA4(I), I
     1=1, J)
  91  FORMAT(5X, 11F10.5)
C
```

```
      NS = 2
C
C     IF READ IN ANGLES ARE DONE BY INCREMENTS NO = 2 AND DO LOOP IS
C     EXITED, OTHERWISE A NEW ANGLE IS READ IN
      GO TO (33, 17),NO
C
C     IF ENERGIES ARE DONE BY INCREMENTS NOT = 2 AND DO LOOP IS EXITED,
C     OTHERWISE NEW VALUE OF EINI IS READ IN
   25 GO TO (19,23) NOT
   33 CONTINUE
   19 CONTINUE
   13 CONTINUE
    6 GO TO 50
      END
```

References

1) W. S. C. Williams, An Introduction to Elementary Particles (Academic Press, New York, 1961).

2) A. M. Baldin, V. I. Gol'danskii, and I. L. Rozenthal, Kinematics of Nuclear Reactions (Pergamon Press, New York, 1961).

3) J. D. Jackson, Classical Electrodynamics (John Wiley and Sons, New York, 1962).

ORNL-4179
UC-34 — Physics

INTERNAL DISTRIBUTION

1. Biology Library
2-4. Central Research Library
5-6. ORNL — Y-12 Technical Library
Document Reference Section
7-56. Laboratory Records Department
57. Laboratory Records, ORNL R.C.
58. T. W. Armstrong
59. J. B. Ball
60. H. W. Bertini
61. R. L. Ferguson
62. J. L. Fowler
63. R. L. Hahn
64. H. H. Hubbell
65. M. L. Halbert
66. Troyce Jones
67. G. D. Kerr
68. C. E. Larson
69. F. K. McGowan
70. H. G. MacPherson
71-80. R. E. Phillips
81. J. K. Poggenburg
82. E. Ricci
83. S. J. Rimshaw
84. R. T. Santoro
85. M. J. Skinner
86. A. H. Snell
87-111. S. T. Thornton
112. D. R. Inglis (consultant)
113. T. Lauritsen (consultant)
114. H. W. Newson (consultant)
115. Lloyd Smith (consultant)

EXTERNAL DISTRIBUTION

116. J. A. Swartout, Union Carbide Corporation, New York, N.Y.
117. Laboratory and University Division, AEC, ORO
118-431. Given distribution as shown in TID-4500 under Physics category
(25 copies — CFSTI)